
1. Introduction





   Programs in Use informs you when programs are active within the call stack of open jobs.  There are many situations when you would want to know this, but unfortunately OS/400 provides no easy way to find this out… called programs are not locked so you are not able to tell when a program is being used or which jobs are using it.





   PIU is accessed via a command which allows the entry of generic program and library names.  A list of jobs which have those programs in their call stack is displayed on a screen which shows the five highest (most recent) entries in a call stack beside each job.  The list may be sorted and selected by job name or user.





   This tool may also be accessed as Application Program Interface (API) programs and called from your application programs.  There are two APIs available:





API #1 – Determine program activity status.


This API will pass back information on whether a program is being used.  It can return either a Yes / No flag to say whether a program is active, or a list of jobs which are using that program.





API #2 – Retrieve call stack.


This API will retrieve the call stack of any active job.  This sought-after function is strangely absent from OS/400.  A qualified job name (name + user + number) is sent to the API and the call stack is returned in an array (no user spaces required).  These APIs work at all security levels.


�



2.1 How to Install – FTP method





Note: FTP is the easiest way to install this product.  If you don’t have an FTP connection with your AS/400, see the next section, ‘Installing Using non-FTP methods’





1. On the AS/400 create a save file, e.g.  CRTSAVF QGPL/SAVF2





2.  On your PC start a DOS session, then transfer PIU_R.SAV (for RISC) or PIU_C.SAV (for CISC) to the AS/400 using the following FTP commands :





cd c:\temp                   (or the directory where piu_r.sav resides)


ftp sysname                (where sysname = your AS400 System Name – see footnote)


                                    (enter user id, then password, as prompted by ftp)


bin                               (selects binary image transfer)


put piu_r.sav qgpl/savf2 replace      (copies file to AS400 save file)


quit                                                      (exit ftp)


exit                                                      (exit DOS session)





3.  On the AS/400, check that the save file contains information:


	DSPSAVF SAVF2





4.  Restore the product components:


	RSTOBJ *ALL PIU *SAVF SAVF(SAVF2) RSTLIB(yourlib)


   where yourlib is your utility / programming tools library





      End of procedure.


      (Continue with section ‘Operation – Display Mode’.)





Footnote:  If you don’t know your ‘system name’ you can instead use the direct TCP/IP internet address of your AS/400.  Do a CFGTCP then take option 1, then select an address which has a line type *ELAN.  So the above ftp command would have the form of e.g. ftp 192.168.10.30

2.2 Installing Using non-FTP methods





Use one of these methods if you don’t have an FTP connection with your AS/400.





1.  Create a physical file to receive the save file, e.g.:


	CRTPF QGPL/SAVF1 RCDLEN(528)





2.  Now for the tricky step.  You need to download the PC document PIU_R.SAV (for RISC) or PIU_C.SAV (for CISC) into this physical file.  There are many ways of doing this such as PC Support, RUMBA, Client Access, FTP, and Shared folders.  You may need to have an expert on hand to help with the finer details.


   If you are using something like PC Support, RUMBA or Client Access you will need an ‘.FDF’ template (PC-object) to refer to.  To create one, add a dummy record to the above physical file and upload it to your PC.  This process will create an .FDF object that can be referred to when downloading.  Remember to clear the physical file before doing the download.  Specify the ‘No Conversion’ option on data mapping.


   If  you are using shared folders move the PIU_R.SAV object into a shared folder then enter the command:


	CPYFRMPCD your-folder-name QGPL/SAVF1 PIU_R.SAV


			TRNTBL(*NONE) TRNFMT(*NOTEXT)





3.  Create a save file to receive the physical file, e.g.:


	CRTSAVF QTEMP/SAVF2





4.  Now you must copy the physical file into the save file.  To do this write the below 8-line RPG program and compile it.  This program has the physical file as input and the save file as output.  An array is used to hold the data because the buffer size is too large for a single field.





Source for member: CPYSAVF


        ....+... 1 ...+... 2 ...+... 3 ...+... 4 ...+... 5 ...+... 6 


0001.00      FSAVF1   IP  F     528            DISK                  


0002.00      FSAVF2   O   F     528            DISK                  


0003.00      E                    ARY       528  1                   


0004.00      ISAVF1   AA                                             


0005.00      I                                        1 528 ARY      


0006.00      C                     EXCPT                             


0007.00      OSAVF2   E                                              


0008.00      O                         ARY                           





5.  Now run the program:


	CALL CPYSAVF





6.  Now continue with step 3 in the previous section ‘How to Install – FTP method’





-------------------------------------------





2.3 How to Uninstall





   To uninstall this product, simply delete its components in the yourlib library that you installed it into:





Object		Type


WRKPGINU	*CMD


WRKPGIU	*PGM


WRKPGIUD	*FILE


WRKPGIUR	*PGM


RTVCALST	*PGM





2.4 Installation Library





  This product is shipped in a library called PIU which is intended only as a temporary library.  Normally you would want to move the objects into an existing library already present in your library list, e.g. QGPL, because the objects need to be in the library list.  But if you prefer to leave them in the PUI library, a good idea would be to change the WRKPGMINU command to have PIU as its product library:


CHGCMD WRKPGMINU PRDLIB(PIU)


Then duplicate the command into QGPL.





2.5  APIs as ‘Stand Alone’ Programs





The APIs supplied with this product (WRKPGIU and RTVCALST) can exist by themselves without the need for any other objects in the product library.  These APIs  can be moved into separate libraries and the other PIU objects deleted.  This is possible because the product keys are stored within the API programs, so each program can check it’s key internally.

3.  Operation – Display mode





This tool is activated in display mode by use of the command WRKPGMINU:





The WRKPGMINU – (Work With Programs In Use) command





  The command has a single qualified parameter as follows:





WRKPGMINU     PGM        Program      Program to look for


                         + Library    Library to look for





Program to look for:  Enter the name of a program to be inquired upon as being active.  This can be an actual program name or generic program name.  The following values are supported:





Name		Check for a program of a specific name


Generic*	Check for programs having a certain prefix 


*ALL		Check for all programs in the specified library.  This is the default.





Library to look for:  Enter the name of a program library to be inquired upon as containing the active programs.  This can be an actual library name or generic library name.  The following values are supported:





Name		Look only in a library of a specific name


Generic*	Look  in all libraries having a certain prefix 


*ALL		Look in all libraries.  This is the default.


*ALLUSR	Look in all ‘user libraries’.  A ‘user library’ has a different meaning here from the definition used in OS/400 commands.  It is defined here as being any library which isn’t one of the following:  QSYS, QSYS2, QIWS, QTCP, QPDA, QSQL, QUMEDIA, QHTTPSVR.  This is the default value and is used to filter out OS/400 programs from the display.  Note that the library QQRYLIB (Query/400 library) has been labeled as a user library because it is commonly used by end-users.





*USRLIBL	Look for programs which exist in a library that is part of the ‘user library list’ for the job executing the command.  The ‘user library list’ has a different meaning here from the definition used in OS/400 commands.  It is defined here as being the Product Libraries section of the library list (if any) plus the Current Library (if any) plus the standard User Library List.  I.e. the portion of your (full) library list, excluding the System Library list. This is used to only display programs that form part of an certain application.





Examples





WRKPGMINU PRODOBJ/ABC123


This will list all the jobs running program ABC123 in library PRODOBJ.





WRKPGMINU


This will display a list of all the users on your system and what programs they are using.  Note that this uses the default entries of *ALLUSR/*ALL.





WRKPGMINU *ALL/*ALL


The output of this looks a bit like the program information available in WRKACTJOB except that (a) It includes all active jobs, including those which are in SysReq-1 state or have been disconnected.  (b) It shows the call stack in reverse order, i.e. most recent program first.





WRKPGMINU INV*


This will show all usage of any inventory programs (assuming your inventory programs all start with ‘INV’).  This is equivalent to specifying *ALLUSR/INV*.





WRKPGMINU *USRLIBL/*ALL


This will show the usage of any program that is in your user library list (see above definition of the ‘user library list’ used here.  This is intended to give you a picture of who is signed on to an application.  So to make best use of this option, sign on as a production user before running the above.





The Display





The below display shows a typical output after typing WRKPGMINU, using its default parameters of  all-user-library programs:








                            Work with Programs In Use                   SYSNAME       


                                                              01/02/99  12:34:56


  Program:  *ALL         Library:  *ALLUSR





   4=End   5=Work with   8=Work with spooled files


      Position -> __________


  Opt  Job        User        Pgm 1     Pgm 2     Pgm 3     Pgm 4     Pgm 5


   _   DSP21      ADAM        INV100    MENUPGM


   _   DSP38      BENNY       DRP200    DRP200C   MENUPGM


   _   DRPRUN     BENNY       DRP300           


   _   DSP57      CHRISTINE   INV100    MENUPGM


   _   DSP06      DAVID       MENUPGM         


   _   DSP17      ELIZABETH   MENUPGM         


   _   DSP11      FRED        MENUPGM         


   _   DSP63      GARRY2      ORD510    ORD501    ORD500    MENUPGM


   _   DSP15      HELEN       MENUPGM         


   _   DSP16      IRIS        MENUPGM         


   _   ENDMONTH   QSYSOPR     INV904    INV900    SYS901    SYS900    SYS900C


   _   DSP01      QSYSOPR     OPERMENU        


   _   PC100D1    SIMON       MENUPGM   INV100    MENUPGM


   _   PC100D2    SIMON       INV150    ORD123    INV150    MENUPGM


                                                                         More...


  F3=Exit   F5=Refresh   F11=View 2   F12=Cancel   F16=Sort by Job








In its default mode the display shows a list of interactive and batch jobs which were running the selected programs.  Beside each job is a list of the top five programs that the job had – from the most recent program and in reverse order.  From the above display we can see that some users are sitting on the menu program while others are using menu options.  There are also two batch jobs displayed.





Sorting


The list is default sorted by user name.  Pressing F16 will change the sort to be by job name.  At the top of the user and job name is an input field which allows you to reposition the list.





Views


There are two different view types:


(1) By program – this shows the top five programs in the job stack


(2) By program and library – this shows the program and library name for the top two programs in the job stack.


Pressing F11 will switch you between these views.





Options


Beside the displayed job you can put job options:


4	End job.  This will display the ENDJOB prompt screen first.


5	Work with job


8	Work with job’s spooled files





�Function keys


F3	Exit the program


F5	Refresh the screen


F11	Toggle between the program and the program / library view


F12	Exit the program


F16	Toggle between sorting by job name and by user


Enter	Process options.  Exit if no options.


�
4.  API #1 – Determine program activity status (WRKPGIU)





The main processing program WRKPGIU may be called as an API to determine whether a program is being used.  There are two methods of calling this API: (a) Simple format - which passes back a Yes / No flag to say whether a program is being used, and (b) Extended format - which passes back a list of jobs which are using that program.





4.1 Simple mode





In the simple API mode you can call program WRKPGIU and get it to return a flag saying whether the program was active or not.





Program to call:	WRKPGIU


Parameters:


Program name		10  A	input


Library name		10  A	input


Return flag		  1  A	output





Program name – 10 alpha - the name of the program to check.  This program must be an existing program.  It cannot be a generic value.





Library name – 10 alpha - the name of the library in which the above program resides.  This program must be an existing library.  It cannot be a generic value.  The only special value allowed for this is ‘*LIBL’





Return flag – 1 alpha – this tells you whether the program is being used or not.  Here are the possible return values:





Y	Yes	The program is being run by at least one job in the system.


N	No	The program is not being run by any job on the system, or was rejected


		by the demo version of the product.


E	Error	An error has occurred – see below note 1.


M	Maybe	The program might be being used – see below note 2.





Note 1.  There are a number of reasons why an error may occur, but most likely it’s because the program or library couldn’t be found.  The actual error will be listed in the job log.  This tool is not designed to do error reporting.  If you believe an error is likely to occur,  you should do a CHKOBJ against the program first.





Note 2.  What has happened here is that the API could not find any jobs using the program but there was a job who’s call stack it couldn’t retrieve.  This error normally happens when the job in question is changing it’s call stack too quickly and the API is unable to determine what it is.  (You may sometimes see this problem occurring when displaying a call stack.  An error ‘complete job information not available’ (or similar) gets displayed)  When this problem occurs the API will delay for a second and then try again.  It will retry 5 times at 1 second intervals before giving up and returning the M (maybe) flag or a ‘?’ character for the extended API mode (section 4.2).


   This problem is quite rare and if it does happen it is unlikely that the job in question was running the specified program, so it is probably safe to treat this as a ‘No’.  However you could put some code in your program to re-call the API again.





Example:





The below CL program checks to see if the ‘Invoice Run’ is being used, and returns an error to the user if it is.  In this example BIL900 represents the invoicing program.





PGM


   DCL &INUSE *CHAR 1


   .


   .


   .


   CALL WRKPGIU (‘BIL900’ ‘*LIBL’ &INUSE)


   IF (&INUSE = ‘Y’) DO


      SNDPGMMSG ‘The invoicing program is in use’ +


              TOPGM(*SAME)


      GOTO REDISPLAY


   ENDDO 


   .


   .


   .


ENDPGM








4.2  Extended mode





In extended mode – the API call can return a list of jobs which are using a specified program.  This may be useful if you want to send a message to a user to tell them who is using a program.





Program to call:	WRKPGIU


Parameters:


Program name			10  A	input


Library name			10  A	input


Array elements provided	5,0 P    input 


Array elements available	5,0 P    output 


Return array			27 A * n   output





Program name – 10 alpha - the name of the program to check.  This program must be an existing program.  It cannot be a generic value.





Library name – 10 alpha - the name of the library in which the above program resides.  This program must be an existing library.  It cannot be a generic value.  The only special value allowed for this is ‘*LIBL’





Array elements provided – 5,0 packed – the number of array elements available in the Return array.  The API won’t return any more beyond this number.





Array elements returned – 5,0 packed – the number of array elements available to be returned.  This value may actually be larger than the ‘Array elements provided’ (but the API won’t return any more elements than are provided) – it means that if your array were bigger then this is the number that could be returned.





Return array – 27 alpha times the number of elements you provided – this gives a list of jobs that are using this program.  Each array element has the following layout:





From	To	Field name


  1	10	Job name


11	20	User name


21	26	Job number


27	27	Error flag





Job name – 10 alpha – the name of the job using the program


User name – 10 alpha – the name of the user of the job using the program


Job number – 6 alpha – the job number of the job using the program


Error flag – 1 alpha – this indicates whether there was an error occurred when 


processing the job.  This has the following values:


		Blank	- no error occurred, this job is using the program


		?	- an error occurred when processing the job.  See above Note 2, 


  sect. 4.1.





Example 1:





The below CL program checks to see if the ‘Invoice Run’ is being used, and if it is, it returns an error to the user giving the name of the first user that is running it.  In this example BIL900 represents the invoicing program.  The error flag isn’t checked for.





PGM


   DCL &PROV *DEC (5 0)


   DCL &AVAL *DEC (5 0)


   DCL &RTNJOB *CHAR 27


   DCL &USERNAM *CHAR 10


   .


   .


   .


   CHGVAR &PROV 1


   CALL WRKPGIU (‘BIL900’ ‘*LIBL’ &PROV &AVAL &RTNJOB)


   IF (&AVAL > 0) DO


      CHGVAR &USERNAM %SST(&RTNJOB 11 10)


      SNDPGMMSG (‘The invoicing program is being used by’ +


              *BCAT &USERNAM *TCAT ‘.’) TOPGM(*SAME)


      GOTO REDISPLAY


   ENDDO 


   .


   .


   .


ENDPGM





Example 2:





In this example an RPG program is used to get a list of jobs that are running program ABC123 in library PRODOBJ into an array for further processing.  The error flag will be checked to make sure it’s not set.





....+... 1 ...+... 2 ...+... 3 ...+... 4 ...+... 5 ...+... 6 ...+


      *                                                      


      *  Array to hold returned data                         


     E                    RTN       100 27                   


      *                                                      


      *  Data structure to split each array element          


     IRTNDS       DS                                         


     I                                        1  10 RJOB     


     I                                       11  20 RUSER    


     I                                       21  26 RJNO     


     I                                       27  27 RERR     


      *                                                      


      *  Call API for first 100 jobs running program         


     C                     CALL ‘WRKPGIU’                    


     C                     PARM ‘ABC123’  PROG   10          


     C                     PARM ‘PRODOBJ’ LIBR   10          


     C                     PARM 100       PROV    50         


     C                     PARM           AVAL    50         


     C                     PARM           RTN                


      *                                                      


      *  Trim available if > provided                        


     C           AVAL      IFGT PROV                         


     C                     Z-ADDPROV      AVAL               


     C                     ENDIF                             


      *                                                      


      *  Read through all returned array elements            


     C           1         DO   AVAL      X       50         


     C                     MOVE RTN,X     RTNDS              


      *                                                      


      *  Ignore error jobs                                   


     C           RERR      IFNE ‘?’                          


      *                                                      


      *    Put desired processing here!                      


      *                                                      


     C                     ENDIF                             


     C                     ENDDO                             


      *                                                      


     C                     SETON                     LR      








5.  API #2 – Retrieve Call Stack (RTVCALST)





This API retrieves the call stack of a given job.  Its input is a qualified job name and its output is an array of program names and libraries that are currently running in the stack.





Program to call: 	RTVCALST





Parameters:


Qualified job name		26 A	input


Format			  	3 A	input


Array elements provided	5,0 P	input 


Array elements available	5,0 P	output 


Return array			20 A * n   output





Qualified job name - 26 alpha - the job that will have it's stack retrieved.  This field is made up as follows:


 From  To  Field


      1 - 10   Job name


    11 - 20   Job user


    21 - 26   Job number





Special case: If position 1 of qualified job name is an '*' (asterisk), then the current job will have its stack retrieved.





Format - 3 alpha - determines what format the information will be returned in.  There are 2 possible values for this:


100 - Forward order


110 - Backward order


   In Forward order, the first element of the return array will be the earliest called program and the last element will be the most recently called program that the array can hold.


   In Backward order, the first element of the return array will be the most recently called program, i.e. the program that called this API, and the last element will be the earliest called program that the array can hold.


   The advantage of Backward order is that it allows you to easily determine the name & library of a program, and what called it, using a small return array size (see example 4 below).





Array elements provided - 5,0 packed - the number of array elements available in the Return array.  The API won't return any more beyond this number.





Array elements returned - 5,0 packed - the number of array elements available to be returned.  This value may actually be larger than the 'Array elements provided' (but the API won't return any more elements than are provided) - it means that if your array were bigger then this is the number that could be returned.





Return array - 20 alpha times the number of elements you provided - this gives a list of jobs that are using this program.  Each array element has the following layout:





From	To	Field name


  1	10	Program name


 11	20	Library name





Note:  This API will not be included in the return array, even though it was present in the call stack at the time of retrieval.





Error conditions:


If any errors are encountered, the API will return a value of one in the Array Elements Available parameter.  The first element of the return array will hold an error condition code.  The error codes (returned in the first 7 characters of element 1) are:


  *BADFMT  - Invalid format code (parameter 2)


  *NOTFND  - Job not found


  *NOTACT  - Job not active


  *NOTAUT  - Not authorized to foreign job





Limitations:


The API can return a maximum of 256 program and library names.  If there are more programs than this in the stack, the first 256 names will be returned and additional (most recently called) programs will be ignored.





Performance:


The program runs at its fastest when retrieving the current job stack - when job name = '*'.  The program also runs quickly when getting the stack of a foreign job multiple times.  This is because the program stays open and will not need to 'look' for the job if it is the same as the previous job.





Example 3:


The below example is an RPG program.  Its input is a Job name, User and Number.  It retrieves the call stack for that job and processes each entry in the 'PUT DESIRED PROCESSING HERE!' section.





....+... 1 ...+... 2 ...+... 3 ...+... 4 ...+... 5 ...+... 6 ...+


      *


      *  ARRAY TO HOLD RETURNED DATA


     E                    RTN       100 20


      *


      *  DATA STRUCTURE TO HOLD QUALIFIED JOB NAME


     IJOBQAL      DS


     I                                        1  10 JOBNAM


     I                                       11  20 JOBUSR


     I                                       21  26 JOBNUM


      *


      *  DATA STRUCTURE TO SPLIT EACH ARRAY ELEMENT


     IRTNDS       DS


     I                                        1  10 RPGM


     I                                       11  20 RLIB


      *


     C           *ENTRY    PLIST


     C                     PARM           ENTJOB 10


     C                     PARM           ENTUSR 10


     C                     PARM           ENTJBN  6


      *


      *  PRIME QUALIFIED JOB NAME


     C                     MOVE ENTJOB    JOBNAM


     C                     MOVE ENTUSR    JOBUSR


     C                     MOVE ENTJBN    JOBNUM


      *


      *  CALL API FOR FIRST 100 PROGRAMS IN JOB STACK


     C                     CALL 'RTVCALST'


     C                     PARM           JOBQAL


     C                     PARM '100'     FORMAT  3


     C                     PARM 100       PROV    50


     C                     PARM           AVAL    50


     C                     PARM           RTN


      *


      *  TRIM AVAILABLE IF > PROVIDED


     C           AVAL      IFGT PROV


     C                     Z-ADDPROV      AVAL


     C                     ENDIF


      *


      *  READ THROUGH ALL RETURNED ARRAY ELEMENTS


     C           1         DO   AVAL      X       50


     C                     MOVE RTN,X     RTNDS


      *


      *    PUT DESIRED PROCESSING HERE!


      *


     C                     ENDDO


      *


     C                     SETON                     LR





Example 4:


The below example is a CL program.  It calls the API in Backward order to retrieve the 2 most recent entries of the call stack for the current job.  It then extracts the program and library names of the current program and calling program and displays them.  By using the Backward format, the return array can be a fixed 40 character field – so there is no need for array processing.





....+... 1 ...+... 2 ...+... 3 ...+... 4 ...+... 5 ...+... 6 ...+... 7


             PGM





             DCL        VAR(&PROV) TYPE(*DEC) LEN(5 0) VALUE(2)


             DCL        VAR(&AVAL) TYPE(*DEC) LEN(5 0)


             DCL        VAR(&RTN) TYPE(*CHAR) LEN(40)


             DCL        VAR(&THSPGM) TYPE(*CHAR) LEN(10)


             DCL        VAR(&THSLIB) TYPE(*CHAR) LEN(10)


             DCL        VAR(&PRVPGM) TYPE(*CHAR) LEN(10)


             DCL        VAR(&PRVLIB) TYPE(*CHAR) LEN(10)





             CALL       PGM(RTVCALST) PARM('*' '110' &PROV &AVAL &RTN)





             CHGVAR     VAR(&THSPGM) VALUE(%SST(&RTN 01 10))


             CHGVAR     VAR(&THSLIB) VALUE(%SST(&RTN 11 10))


             CHGVAR     VAR(&PRVPGM) VALUE(%SST(&RTN 21 10))


             CHGVAR     VAR(&PRVLIB) VALUE(%SST(&RTN 31 10))





             SNDPGMMSG  MSG('This program is ' *CAT &THSLIB *TCAT +


                          '/' *CAT &THSPGM)


             SNDPGMMSG  MSG('Calling program is ' *CAT &PRVLIB *TCAT +


                          '/' *CAT &PRVPGM)





             ENDPGM





Example output when calling above program:





   This program is QTEMP/TEST2


   Calling program is QSYS/QUOCMD


�



6.  Legal information





Conditions of use





To legally use this product you must register it. Without registration, this product may not be used beyond the date allowed for by a time-trial security code supplied by Precosis Pty Ltd, unless it is used in a manner allowed for by the demonstration version of the product. With registration it may only be used as a permanent version on the machines to which it is registered.





To Register





See the attached emailed or Internet document for details on registering this product.





Copyrights





This package and its components and these web pages and their components are the Copyright of Precosis Pty Ltd 1999-2000. All rights reserved. 





Disclaimer





All reasonable efforts have been made to ensure that this product works in the manner described but neither Precosis Pty Ltd nor the authors of this product guarantee that the product will work in this manner. No liability will be accepted for any damage arising from the use of this product.

